# Maintainer: Alexey Pavlov <alexpux@gmail.com>

pkgname=windows-default-manifest
pkgver=6.4
pkgrel=1
pkgdesc='Default Windows application manifest'
url='https://cygwin.com/'
arch=('i686' 'x86_64')
license=('BSD')
makedepends=('git' 'gcc')
_GIT_TAG="release-${pkgver//\./_}"
source=("git://sourceware.org/git/cygwin-apps/${pkgname}.git#tag=${_GIT_TAG}"
        001-msysize.patch)
sha256sums=('SKIP'
            '0baef1e5f9b980cdf730386a652b6b02ba6b675cc120ad67c7c15e6a88050e0f')

prepare() {
  cd "${pkgname}"
  patch -p1 -i ${srcdir}/001-msysize.patch
  autoreconf -fiv
}

build() {
  cd ${srcdir}/${pkgname}
  ./configure --prefix=/usr
  make
}

package() {
  cd ${srcdir}/${pkgname}
  make DESTDIR=${pkgdir} install
}
